[bookmark: _GoBack]TITLE: LCSM model for Objective Memory and Subjective Memory;
DATA: FILE is C:\Users\ …..\Dataset.dat;
VARIABLE:NAMES ARE RESEARCHID sm_c1 sm_c2 sm_c3 sm_c4 sm_c5 om_c1 om_c2 om_c3       om_c4 om_c5 oe_c1 oe_c2 oe_c3 oe_c4 oe_c5 ol_c1 ol_c2 ol_c3 ol_c4 ol_c5 depres age edugp gender;
     Usevariables = age gender  depres  x1 x2 x3 x4 x5 y1 y2 y3 y4 y5 edugp_1 edugp_2;
    Missing are ALL(9999);
DEFINE:
            x1 = sm_c1;
            x2 = sm_c2;
            x3 = sm_c3;
            x4 = sm_c4;
            x5 = sm_c5;
            y1 = om_c1;
            y2 = om_c2;
            y3 = om_c3;
            y4 = om_c4;
            y5 = om_c5;
            edugp_1 = 0;
            if(edugp_re eq 1)then edugp_1 = 1;
            edugp_2 = 0;
            if(edugp_re eq 2)then edugp_2 = 1;    

ANALYSIS: TYPE=MEANSTRUCTURE;
                      ITERATIONS = 8000;
MODEL:
!For the variable y
!Oberved variables and latent level
ly1 by y1 @1;
ly2 by y2 @1;
ly3 by y3 @1;
ly4 by y4 @1;
ly5 by y5 @1;

!Autoregressive part
ly2 on ly1 @1;
ly3 on ly2 @1;
ly4 on ly3 @1;
ly5 on ly4 @1;

!Difference score on latent level
dy1 by ly2 @1;
dy2 by ly3 @1;
dy3 by ly4 @1;
dy4 by ly5 @1;

!Auto-porpotion of difference score on level
! add the starting values after *
dy1 on ly1 * (1);
dy2 on ly2 * (1);
dy3 on ly3 * (1);
dy4 on ly4 * (1);

!Model relationship between slope and ds
sy by dy1 @1;
sy by dy2 @1;
sy by dy3 @1;
sy by dy4 @1;
y0 by ly1 @1;

! Set the means and variance to be 0
[y1@0]; [ly1@0]; [dy1@0]; ly1@0; dy1@0;
[y2@0]; [ly2@0]; [dy2@0]; ly2@0; dy2@0;
[y3@0]; [ly3@0]; [dy3@0]; ly3@0; dy3@0;
[y4@0]; [ly4@0]; [dy4@0]; ly4@0; dy4@0;
[y5@0]; [ly5@0];ly5@0;

!beginning the codes for x variables
!For the variable x
!Oberved variables and latent level
lx1 by x1 @1;
lx2 by x2 @1;
lx3 by x3 @1;
lx4 by x4 @1;
lx5 by x5 @1;

!Autoregressive part
lx2 on lx1 @1;
lx3 on lx2 @1;
lx4 on lx3 @1;
lx5 on lx4 @1;

!Difference score on latent level
dx1 by lx2 @1;
dx2 by lx3 @1;
dx3 by lx4 @1;
dx4 by lx5 @1;


!Auto-porpotion of difference score on level
! add the starting values after *
dx1 on lx1 * (2);
dx2 on lx2 * (2);
dx3 on lx3 * (2);
dx4 on lx4 * (2);

!Model relationship between slope and ds
sx by dx1 @1;
sx by dx2 @1;
sx by dx3 @1;
sx by dx4 @1;
x0 by lx1 @1;

 !Set the means and variance to be 0
[x1@0]; [lx1@0]; [dx1@0]; lx1@0; dx1@0;
[x2@0]; [lx2@0]; [dx2@0]; lx2@0; dx2@0;
[x3@0]; [lx3@0]; [dx3@0]; lx3@0; dx3@0;
[x4@0]; [lx4@0]; [dx4@0]; lx4@0; dx4@0;
[x5@0]; [lx5@0];lx5@0;

!Covariates on the Intercept;
x0 on age *;
x0 on edugp_1* ;
x0 on edugp_2* ;
x0 on gender*;
x0 on depres;

y0 on age *;
y0 on edugp_1* ;
y0 on edugp_2* ;
y0 on gender*;
y0 on depres;

!Covariates on the Slope;
sx on age *;
sx on edugp_1* ;
sx on edugp_2* ;
sx on gender*;
sx on depress*;

sy on age *;
sy on edugp_1*;
sy on edugp_2* ;
sy on gender*;
sy on depress*;

!coupling from OM to SM
dx1 on ly1*(3);
dx2 on ly2*(3);
dx3 on ly3*(3);
dx4 on ly4*(3);

!coupling from SM to OM
dy1 on lx1*(4);
dy2 on lx2*(4);
dy3 on lx3*(4);
dy4 on lx4*(4);

[y0 x0 sy sx];
y0 x0 sy sx;

!Set all residuals to be equal
y1* (5);x1* (6);
y2* (5);x2* (6);
y3* (5);x3* (6);
y4* (5);x4* (6);
y5* (5);x5* (6);

! Set the correlated residuals
y1 with x1* (7);
y2 with x2* (7);
y3 with x3* (7);
y4 with x4* (7);
y5 with x5* (7);

OUTPUT:  sampstat standardized;
